

[Description](#)
[Syntax](#)
[Remarks and examples](#)
[Also see](#)

Description

`estimates` allows you to store and manipulate estimation results:

- You can save estimation results in a file for use in later sessions.
- You can store estimation results in memory so that you can
 - a. switch among separate estimation results and
 - b. form tables combining separate estimation results.

Syntax

Command	Reference
<i>Save and use results from disk</i>	
<code><u>estimates</u> save <i>filename</i></code>	[R] estimates save
<code><u>estimates</u> use <i>filename</i></code>	[R] estimates save
<code><u>estimates</u> <u>describe</u> using <i>filename</i></code>	[R] estimates describe
<code><u>estimates</u> <u>esample</u>: ...</code>	[R] estimates save
<i>Store and restore estimates in memory</i>	
<code><u>estimates</u> <u>store</u> <i>name</i></code>	[R] estimates store
<code><u>estimates</u> <u>restore</u> <i>name</i></code>	[R] estimates store
<code><u>estimates</u> <u>query</u></code>	[R] estimates store
<code><u>estimates</u> <u>dir</u></code>	[R] estimates store
<code><u>estimates</u> <u>drop</u> <i>namelist</i></code>	[R] estimates store
<code><u>estimates</u> <u>clear</u></code>	[R] estimates store
<i>Set titles and notes</i>	
<code><u>estimates</u> <u>title</u>: <i>text</i></code>	[R] estimates title
<code><u>estimates</u> <u>title</u></code>	[R] estimates title
<code><u>estimates</u> <u>notes</u>: <i>text</i></code>	[R] estimates notes
<code><u>estimates</u> <u>notes</u></code>	[R] estimates notes
<code><u>estimates</u> <u>notes</u> <u>list</u> ...</code>	[R] estimates notes
<code><u>estimates</u> <u>notes</u> <u>drop</u> ...</code>	[R] estimates notes
<i>Report</i>	
<code><u>estimates</u> <u>describe</u> [<i>name</i>]</code>	[R] estimates describe
<code><u>estimates</u> <u>replay</u> [<i>namelist</i>]</code>	[R] estimates replay

Command, continued	Reference, continued
<i>Tables and statistics</i>	
<code>estimates table [namelist]</code>	[R] estimates table
<code>etable</code>	[R] etable
<code>estimates selected [namelist]</code>	[R] estimates selected
<code>estimates stats [namelist]</code>	[R] estimates stats
<code>estimates for namelist: ...</code>	[R] estimates for

Remarks and examples

`estimates` is for use after you have fit a model, be it with `regress`, `logistic`, etc. You can use `estimates` after any estimation command, whether it be an official estimation command of Stata or a community-contributed one.

`estimates` has three separate but related capabilities:

1. You can save estimation results in a file on disk so that you can use them later, even in a different Stata session.
2. You can store up to 300 estimation results in memory so that they are at your fingertips.
3. You can make tables comparing any results you have stored in memory.

Remarks are presented under the following headings:

[Saving and using estimation results](#)
[Storing and restoring estimation results](#)
[Comparing estimation results](#)
[Jargon](#)

Saving and using estimation results

After you have fit a model, say, with `regress`, type

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)
. regress mpg weight displ foreign
(output omitted)
```

You can save the results in a file:

```
. estimates save basemodel
file basemodel.ster saved
```

Later, say, in a different session, you can reload those results:

```
. estimates use basemodel
```

The situation is now nearly identical to what it was immediately after you fit the model. You can replay estimation results:

```
. regress
(output omitted)
```

You can perform tests:

```
. test foreign==0
(output omitted)
```

And you can use any postestimation command or postestimation capability of Stata. The only difference is that Stata no longer knows what the estimation sample, `e(sample)` in Stata jargon, was. When you reload the estimation results, you might not even have the original data in memory. That is okay. Stata will know to refuse to calculate anything that can be calculated only on the original estimation sample.

If it is important that you use a postestimation command that can be used only on the original estimation sample, there is a way you can do that. You use the original data and then use `estimates esample:` to tell Stata what the original sample was.

See [R] [estimates save](#) for details.

Storing and restoring estimation results

Storing and restoring estimation results in memory is much like saving them to disk. You type

```
. estimates store base
```

to save the current estimation results under the name `base`, and you type

```
. estimates restore base
```

to get them back later. You can find out what you have stored by typing

```
. estimates dir
```

Saving estimation results to disk is more permanent than storing them in memory, so why would you want merely to store them? The answer is that, once they are stored, you can use other `estimates` commands to produce tables and reports from them.

See [R] [estimates store](#) for details about the `estimates store` and `restore` commands.

Comparing estimation results

Let's say that you have done the following:

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)
. regress mpg weight displ
(output omitted)
. estimates store base
. regress mpg weight displ foreign
(output omitted)
. estimates store alt
```

You can now get a table comparing the coefficients:

```
. estimates table base alt
```

Variable	base	alt
weight	-.00656711	-.00677449
displacement	.00528078	.00192865
foreign		-1.6006312
_cons	40.084522	41.847949

`estimates table` can do much more; see [R] [estimates table](#). `etable` also produces tables from stored estimates, customizes the tables, and exports to a variety of formats, including Word, PDF, L^AT_EX, Excel, and HTML; see [R] [etable](#). Also see [R] [estimates stats](#). `estimates stats` works like `estimates table` but produces model comparisons in terms of BIC and AIC.

Jargon

You know that if you fit a model, say, by typing

```
. regress mpg weight displacement
```

then you can later replay the results by typing

```
. regress
```

and you can do tests and calculate other postestimation statistics by typing

```
. test displacement==0
. estat vif
. predict mpghat
```

As a result, we often refer to the *estimation results* or the *current estimation results* or the *most recent estimation results* or the *last estimation results* or the *estimation results in memory*.

With `estimates store` and `estimates restore`, you can have many estimation results in memory. One set of those—the set most recently estimated or the set most recently restored—is the *current* or *active* estimation results, which you can replay, which you can test, or from which you can calculate postestimation statistics.

Current and *active* are the two words we will use interchangeably from now on.

Also see

[P] [_estimates](#) — Manage estimation results

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).